TP2

Nombres complexes et équations différentielles

1 Gestion des variables

Une **variable** est un nom dans lequel on peut stocker la donnée d'un objet Maple, que ce soit une fonction, une expression ou même une valeur particulière.
Les noms de ces variables pourront être quelconques à l'exception de celles protégées par Maple:

`Nom de la variable`	Description
`D`	Fonction dérivée
`Digits`	Nombre de chiffres significatifs par défaut sur la feuille de travail
`Gamma`	Constante d'Euler
`I`	i complexe
`infinity`	∞
`Pi`	Valeur de π
`cos,sin,...`	le nom des fonctions usuelles (voir TP1)

Pour affecter une nouvelle variable, on utilise le symbole d'affectation `:=`

> `z:=4`

$$z := 4$$

Cependant, il s'agira de rester vigilant car ces affectations sont globales et seront utilisées par Maple dans toute la feuille de travail:

> `P:=a*z^2+b*z+c`

$$P := 16a + 4b + c$$

Pour tester si une variable est affectée, il suffit simplement d'entrer son nom ou bien d'utiliser la commande `assigned`:

> `z; assigned(z); a; assigned(a);`

$$4$$
$$true$$
$$a$$
$$false$$

A la fin d'un exercice, on n'oubliera donc pas de libérer le nom des variables créées grâce aux symboles `''`:

> `z:='z'; assigned(z);`

$$z := z$$
$$false$$

ou plus radicalement en réinitialisant toutes les variables définies sur la feuille de travail:

> `restart`

Si la gestion des variables est très facile, nous serons souvent amenés à les utiliser. En particulier, on s'en servira pour exploiter le résultat d'une commande Maple:

> `restart; s:=solve(z^3=1,z);`

$$s := 1, -\frac{1}{2} + \frac{1}{2}I\sqrt{3}, -\frac{1}{2} - \frac{1}{2}I\sqrt{3}$$

Ce résultat est donné sous la forme d'une *sequence* s dont les éléments peuvent être exploités par leur numéro:

> `s`$_1$`; s`$_2$`; s`$_3$`; s`$_1$`+s`$_2$`+s`$_3$`;`

$$1$$
$$-\frac{1}{2} + \frac{1}{2}I\sqrt{3}$$
$$-\frac{1}{2} - \frac{1}{2}I\sqrt{3}$$
$$0$$

ou bien

> `s[1]; s[2]; s[3]; s[1]+s[2]+s[3];`

$$1$$
$$-\frac{1}{2} + \frac{1}{2}I\sqrt{3}$$
$$-\frac{1}{2} - \frac{1}{2}I\sqrt{3}$$
$$0$$

2 Nombres complexes

Maple permet de faire des calculs dans le corps des complexes en utilisant le symbole `I` pour représenter le i complexe:

> `I`^{2}`; (4+3*I)+(7-2*I); (1+I)*(2-3*I);`

$$-1$$
$$11+I$$
$$5-I$$

On peut isoler les parties réelles et imaginaires avec les commandes `Re` et `Im`:

> `Re(%); Im(%%);`

$$5$$
$$-1$$

On peut prendre le conjugué d'un nombre complexe, calculer son module et son argument:

> `z:=4+3*I: conjugate(z); abs(z); argument(z);`

$$4-3I$$
$$5$$
$$\arctan(\frac{3}{4})$$

Maple est capable de manipuler les complexes sous différentes formes; on peut par exemple définir un nombre complexe sous sa forme exponentielle puis l'écrire sous forme cartésienne, ou inversement:

> `z1:=polar(1,Pi/6); evalc(z1); z2:=1+I; convert(z2,polar);`

$$z1 := polar(1, \frac{1}{6}\pi)$$
$$\frac{1}{2}\sqrt{3}+\frac{1}{2}I$$
$$z2 := 1+I$$
$$polar(\sqrt{2}, \frac{1}{4}\pi)$$

3 Equations différentielles

3.1 Résolution formelle et représentation des solutions

Maple est un outil très puissant; il permet en outre de résoudre formellement des équations différentielles avec la commande `dsolve`, qu'on applique ici à deux exemples bien connus:

E.D. d'ordre 1: $y'(x) - ay(x) = 0$

> `eq1:=diff(y(x),x)-a*y(x)=0`

$$eq1 := \frac{d}{dx}y(x) - ay(x) = 0$$

> `dsolve(eq1,y(x))`

$$y(x) = C1e^{ax}$$

E.D. d'ordre 2: $y"(x) - a^2y(x) = 0$

> `eq2:=diff(y(x),x$2)-a`^{2}`*y(x)=0`

$$eq2 := \frac{d^2}{dx^2}y(x) - a^2y(x) = 0$$

> `dsolve(eq2,y(x))`

$$y(x) = C1e^{-ax} + C2e^{ax}$$

Les constantes $C1$ et $C2$ représentent les constantes d'intégration qui peuvent être déterminées par la donnée de conditions initiales.
D'ailleurs, si celles-ci nous sont indiquées, il suffira de les ajouter à la commande `dsolve` en remplaçant l'équation par l'ensemble suivant: $\{equation, conditions\ initiales\}$

> `dsolve({eq1,y(0)=2},y(x))`

$$y(x) = 2e^{ax}$$

> `dsolve({eq2,y(0)=2,D(y)(0)=0},y(x))`

$$y(x) = e^{-ax} + e^{ax}$$

Dans ces deux cas, Maple renvoie la solution sous la forme d'une égalité $y(x) = ...$. Ainsi, si on veut tracer les courbes représentatives associées, seules les expressions algébriques nous intéressent. On récupère alors la partie de droite de l'égalité (*resp.* de gauche) par la commande `rhs` (*resp.* `lhs`):

> `sol1:=rhs(%)`

$$sol1 := 2e^{ax}$$

> `sol2:=rhs(%)`

$$sol2 := e^{-ax} + e^{ax}$$

Enfin, on rappelle ici comment tracer ces *expressions*: avec $a = 4$,

> `a:=4; plot(sol1,x=-1..1,y=-20..20);`



> `a:=4; plot(sol2,x=-1..1,y=-20..20);`



3.2 Résolution numérique et représentation des solutions

Certaines équations différentielles ne peuvent malheureusement pas être résolues sous une forme explicite:

> `eq:=diff(y(x),x$2)+sin(y(x))=0:   dsolve(eq,y(x));`

$$\int^{y(x)} \frac{1}{\sqrt{2\cos(\_a) + \_C1}} d\_a - x - \_C2 = 0, \int^{y(x)} \left(-\frac{1}{\sqrt{2\cos(\_a) + \_C1}}\right) d\_a - x - \_C2 = 0$$

Cependant, avec des conditions initiales fixées, il existe des méthodes numériques qui donnent la solution ponctuellement. Il suffit d'ajouter l'option `numeric` à la commande `dsolve`. On notera alors `resnum` le résultat:

> `resnum:=dsolve({eq,y(0)=Pi/8,D(y)(0)=0},y(x),numeric)`

$$resnum := \mathbf{proc}(x\_rkf45)\ ...\mathbf{end}$$

La réponse de Maple est surprenante. Il a en fait construit une procédure *resnum* qui à une valeur numérique donnée détermine le point associé sur la courbe solution:

> `resnum(1)`

$$[x = 1., y(x) = 0.215837124424383214, \frac{d}{dx}y(x) = -0.325324298194684226]$$

Cela signifie que la courbe solution passe par le point de coordonnées: $(1, 0.215837124424383214)$; le troisième élément nous donnant la pente de la tangente en ce point.
Ainsi, en itérant ce procédé sur plusieurs valeurs de x, on peut en déduire la solution point par point.

Enfin, si on veut représenter cette solution, on peut définir une *fonction* à partir de ces résultats, mais il s'agit d'être habile.
Plaçons-nous en $x = t$, on sait que $resnum(t)$ renvoie une liste de trois éléments. Seul le deuxième élément nous intéresse pour connaître la valeur de la solution en $x = t$. Celui-ci étant une égalité, il ne faudra considérer que la valeur de droite. Ainsi, on définit la solution cherchée par la fonction:

> `f:=t->rhs(resnum(t)[2])`

$$f := t \to rhs(resnum(t)_2)$$

On trace alors la *fonction* f à la façon des *expressions* mais sans variable:

$>$ `plot(f,-5..5,-1..1)`



Exercice 1 On considère le nombre complexe: $z = (\dfrac{1+i\sqrt{3}}{1-i})^{20}$.

1. Ecrire z sous sa forme cartésienne.
2. En déduire son module et son argument.

Exercice 2 On considère l'équation: $z^6 = 1$.

1. D'après votre cours, préciser les solutions de cette équation.
2. Résoudre cette équation avec Maple. On notera *sol* l'ensemble de ces solutions.
3. Donner les solutions sous forme cartésienne (utiliser la commande `seq`). On appellera *solc* ce nouvel ensemble.
4. Donner les solutions sous forme exponentielle et retrouver votre première réponse (utiliser la commande `seq`). On appellera *solp* ce nouvel ensemble.
5. Calculer la somme des solutions de *solp* (utiliser la commande `sum`).

Exercice 3 On note $\mathcal{E}$ l'ensemble des nombres complexes z tels que: $|z+1-i| = |z+3|$.

1. Décrire l'ensemble $\mathcal{E}$.
2. On pose $z = x + iy$.
   Sous Maple, pour retrouver une équation cartésienne décrivant l'ensemble $\mathcal{E}$, on procèdera en deux temps:
   (a) Transformer l'équation initiale avec la commande `evalc`.
   (b) Résoudre cette nouvelle équation en x <u>et</u> y (utiliser l'aide en ligne).

Exercice 4 On considère l'équation différentielle: $y'(x) + 3y(x) = x^3e^x$ (E).

1. Déterminer l'ensemble des solutions de cette équation.
2. On va maintenant tracer différentes solutions en fonction des conditions initiales.
   (a) Déterminer l'unique solution de (E) vérifiant $y(0) = 2$. On notera $sol1$ son expression.
   (b) Déterminer l'unique solution de (E) vérifiant $y(0) = 0.5$. On notera $sol2$ son expression.
   (c) Déterminer l'unique solution de (E) vérifiant $y(0) = 0$. On notera $sol3$ son expression.
   (d) Déterminer l'unique solution de (E) vérifiant $y(0) = -0.5$. On notera $sol4$ son expression.
   (e) Représenter ces solutions sur un même graphique par la commande:
   `plot([sol1,sol2,sol3,sol4],x=-2..2,y=-5..5)`

Exercice 5 On considère l'équation différentielle: $xy'(x) + 2y(x) = \dfrac{x}{1+x^2}$ (E).

1. Déterminer l'ensemble des solutions définies sur $\mathbb{R}^*$.
2. On appelle h la fonction définie sur $\mathbb{R}^*$ par $h(x) = \dfrac{x - \arctan(x)}{x^2}$.
   (a) Calculer les limites: $\lim_{x\to 0, x\neq 0} h(x)$ et $\lim_{x\to 0, x\neq 0} \dfrac{h(x)}{x}$.
   (b) En considérant l'ensemble des solutions de (E) ci-dessus, déterminer l'unique solution définie sur $\mathbb{R}$. On représentera cette solution pour $x \in [-10; 10]$ et $y \in [-0.5; 0.5]$.

Exercice 6 On considère un pendule simple dont l'équation horaire est définie par:

$$y"(t) + \sin(y(t)) = 0 \quad (eq1)$$

La fonction y désigne l'angle entre l'axe vertical et le pendule.
Pour de petites amplitudes, on lui préfèrera l'équation linéarisée: $y"(t) + y(t) = 0$ $(eq2)$.

1. A $t = 0$, on lâche le pendule sans vitesse: $D(y)(0) = 0$ et avec l'angle: $y(0) = \dfrac{\pi}{8}$.
   (a) Dans ces conditions, déterminer la solution formelle de l'équation linéarisée $(eq2)$. On notera *sol* l'expression de cette solution.
   (b) Dans ces conditions, résoudre numériquement l'équation initiale $(eq1)$: on construira la procédure *resnum*, puis la fonction solution f. On vérifiera que: $f(0) = \dfrac{\pi}{8}$.
   (c) On construit alors la fonction g à partir de l'expression *sol* par la commande: `g:=unapply(sol,t)`.

   Tracer sur un même graphique les fonctions f et g; les abscisses seront comprises entre 0 et 20 et les ordonnées entre -2.5 et 2.5.
2. Mêmes questions en changeant les conditions initiales: $D(y)(0) = 0$ et $y(0) = \dfrac{3\pi}{4}$.

Ainsi, lorsque l'amplitude est trop grande le choix de travailler avec l'équation différentielle linéarisée n'est plus justifiée.